AN EXPLICIT SCHEME FOR THE EQUATION $Q_{ttz} - c^2\gamma Q_{xxz} = c(\beta-\alpha\gamma)Q_{xxt}$.

by Goran C. Starius

Let us consider the equation $\hat{P}_z = \frac{i\omega}{c}\left(1 + \frac{c^2}{\omega^2}\partial_x^2\right)^{1/2}\hat{P}$, where $\hat{P}$ is the Fourier-transform with respect to t of a solution P to the wave equation. By replacing the square root operator by a rational approximation

$$\frac{\alpha + \beta\frac{c^2}{\omega^2}\ \partial_x^2}{1 + \gamma\frac{c^2}{\omega^2}\ \partial_x^2}$$

we get the equation

$$\left(1 + \gamma\frac{c^2}{\omega^2}\partial_x^2\right)\hat{P}_z = \frac{i\omega}{c}\left(\alpha + \beta\frac{c^2}{\omega^2}\ \partial_x^2\right)\hat{P} . \tag{1}$$

After making the variable transformation $\hat{P} = \hat{Q}e^{i\frac{\omega\alpha}{c}z}$ and replacing ω by $i\cdot\partial_t$ we get

$$Q_{ttz} - c^2\gamma Q_{xxz} = c(\beta-\alpha\gamma)\ Q_{xxt} , \tag{2}$$

where Q is the inverse Fourier-transform of $\hat{Q}$. The desired solution $P(x,z,t)$ is simply $Q(x,z,t - \frac{\alpha}{c}z)$ because

$$P(x,z,t) = \int \hat{P}\ e^{-i\omega t}d\omega = \int \hat{Q}e^{-i\omega(t-\frac{\alpha}{c}z)}d\omega = Q(x,z,t - \frac{\alpha}{c}z) .$$

The parameters α,β and γ are chosen such that

$$E(\alpha,\beta,\gamma) = \max\left|\ \sqrt{1-\lambda} - \frac{\alpha-\beta\lambda}{1-\gamma\lambda}\ \right|$$

$$0 \leq \lambda \leq \sin^2\theta$$

is minimized, i.e. the rational approximations we use are optimal in maximum norm, for angles less than θ. In the table below the parameters and the maximal errors are given for different values θ .

TABLE

θ	α	β	γ	min E
45°	0.99973	0.80864	0.31617	2.7×10^{-4}
60°	0.99821	0.85420	0.38323	1.8×10^{-3}
75°	0.99120	0.90836	0.49636	8.8×10^{-3}
85°	0.97448	0.93831	0.62046	2.6×10^{-2}

We shall now derive an explicit difference scheme for equation (2). Let $t_n = n\Delta t$, $z_k = k\Delta z$ and let $Q_k^n(x)$ correspond to $Q(x, z_k, t_n)$ and consider the families of schemes

$$\delta_{tt}(Q_{k+1}^{n+1} - Q_k^{n+1}) + bT[d(Q_{k+1}^{n+2} - Q_k^{n+2}) + (1-2d)(Q_{k+1}^{n+1} - Q_k^{n+1}) + d(Q_{k+1}^{n} - Q_k^{n})] \tag{3}$$

$$+ aT[e(Q_{k+1}^{n+2} - Q_{k+1}^{n+1} + Q_k^{n+1} - Q_k^{n}) + (1-e)(Q_k^{n+2} - Q_k^{n+1} + Q_{k+1}^{n+1} - Q_{k+1}^{n})] = 0$$

$$a = \frac{\Delta t \Delta z c(\beta - \alpha\gamma)}{2(\Delta x)^2}, \qquad b = \frac{(\Delta t)^2 c^2 \gamma}{(\Delta x)^2} .$$

The quantities d and e are arbitrary parameters, δ_{tt} is the usual central difference operator (1, -2, 1) and T is a matrix corresponding to $-\delta_{xx}$. Since we want to solve for Q_{k+1}^{n+2} the scheme is explicit if T is not operating on this quantity. We therefore require that

$$ae + bd = 0 \tag{4}$$

By using this relation in (3) we simply get

$$Q_{k+1}^{n+2} = Q_k^n + [2I - (a+b)T](Q_{k+1}^{n+1} - Q_k^{n+1}) + [I - aT](Q_k^{n+2} - Q_{k+1}^n) \qquad (5)$$

which is independent of both d and e.

Parenthetically we point out the following. Assume that ∂_x^2 in equation (2) is replaced by $-(\Delta x)^{-2}T/(I - \delta T)$, $\delta = 1/12$ for example, instead of just the numerator, then the first term in (3) shall be multiplied by $I - \delta T$ and relation (4) shall be replaced by $ae + bd = \delta$. However, it turns out that we get exactly the same scheme as before, i.e. the one given in (5).

We now turn to a stability investigation. Replace $Q(x,z,t)$ by $Q(z,t)e^{ik_x \cdot x}$ in (5) then we get

$$Q_{k+1}^{n+z} = Q_k^n + 2(1 - 2(a + b))(Q_{k+1}^{n+1} - Q_k^{n+1}) + (1 - 4a)(Q_k^{n+2} - Q_{k+1}^n), \qquad (6)$$

where a and b are those given in (3) multiplied by $\sin^2 \frac{k_x \cdot \Delta x}{2}$.

Let us first consider stability in t-direction, i. e. all terms with index k are considered as inhomogeneous and thus dropped. We get the difference equation

$$Q^{n+2} - 2(1 - 2(a+b))\, Q^{n+1} + (1 - 4a)Q^n = 0$$

whose characteristic equation is

$$\mu^2 - 2(1 - 2(a+b))\mu + 1 - 4a = 0.$$

Thus we must require that $|\mu| \le 1$, which for $a, b \ge 0$ is equivalent to

$$a \le \frac{1-b}{2} . \qquad (7)$$

The overall stability in z-direction can be investigated in the following way. Replace Q_k^n in (6) by $\lambda^k e^{i\omega t_n}$ and we get

$$\lambda e^{i\eta} = e^{-i\eta} + 2(1 - 2(a+b))(\lambda-1) + (1- 4a)(e^{i\eta} - e^{-i\eta}), \ \eta = \omega \Delta t$$

and thus

$$\lambda = \frac{e^{-i\eta} - 2(1-2(a+b)) + e^{i\eta}(1-4a)}{e^{i\eta} - 2(1-2(a+b)) + e^{-i\eta}(1-4a)}$$

Since $|\lambda| = 1$, the only stability condition we get for (5) is the one in (7), where a and b are given in (3).

The scheme was tested for different values of a and b by using Riley's benchmark program (SEP, March 74, page 64). A replacement for Riley's FAST 15 subroutine is attached.

Finally, we point out that if $\alpha = 1$, $\beta = 0.5$ and $\gamma = 0$ then equation (2) is identical to the 15 degree equation (SEP, March 74, page 60).

```
      SUBROUTINE FAST45(Q,NX,NT,A,B,MODE)
      DIMENSION Q(NX,NT),H(240),G(240),F(240),S(240),T(240)
C     ASSUMES THE INPUTS  Q(IX,1)=Q(IX,2)=0 FOR IX=1,NX
      DO 10 IX=1,NX
      H(IX)=0.
10    G(IX)=0.
      NX1=NX-1
      A1=A+B
      A2=2.-2.*A1
      B1=A
      B2=1.-2.*B1
      DO 50 JT=3,NT
      IT=JT
      IF(MODE.EQ.-1)IT=NT+1-JT
      IT1=IT-MODE
      IT2=IT1-MODE
      DO 20 IX=1,NX
      S(IX)=G(IX)-Q(IX,IT1)
20    T(IX)=Q(IX,IT)-H(IX)
      DO 30 IX=2,NX1
30    F(IX)=Q(IX,IT2)+A1*(S(IX-1)+S(IX+1))+
     *A2*S(IX)+B1*(T(IX-1)+T(IX+1))+B2*T(IX)
      F(1)=F(2)
      F(NX)=F(NX1)
      DO 40 IX=1, NX
      Q(IX,IT2)=H(IX)
      H(IX)=G(IX)
40    G(IX)=F(IX)
50    CONTINUE
      DO 60 IX=1,NX
      Q(IX,IT)=G(IX)
60    Q(IX,IT1)=H(IX)
      RETURN
      END
```